
  METHOD if_cod_sls_se_simulate_order~adjust_import_data. 

*-----------------------------------------------------------------------------------------------------* 

* Method        : IF_COD_SLS_SE_SIMULATE_ORDER~ADJUST_IMPORT_DATA 

* Description   : Adding the logic to replace the customer number in incoming data if customer not 

*                 found in S4 because Prospects only exist in C4C not in S4. 

* Dev Scope     : Pricing determination doesn’t work for prospect since prospects are only exist only 

*                 in C4C and not in S4. To achieve the requirement replacing the Customer number in 

*                 the incoming data with default one if the customer number is not found in S4. 

*-----------------------------------------------------------------------------------------------------* 

 

    CONSTANTS : 

      lc_intg_user   TYPE zde_key         VALUE 'C4C_INTG_USER', 

      lc_def_partner TYPE zde_key         VALUE 'DEFAULT_OPPT_QUOTE_PARTNER', 

      lc_cust_role   TYPE but100-rltyp    VALUE 'FLCU01', 

      lc_mesg_cls    TYPE arbgb           VALUE 'ZM_INTG_MESSAGES'. 

 

 

    TYPES : BEGIN OF lty_hardcd, 

              hardcode_key   TYPE zct_avoid_hardcd-hardcode_key, 

              hardcode_value TYPE zct_avoid_hardcd-hardcode_value, 

            END OF lty_hardcd. 

 

    TYPES : BEGIN OF lty_but100, 

              partner TYPE but100-partner, 

              rltyp   TYPE but100-rltyp, 

            END OF lty_but100. 

 

    DATA  : 

      lro_key     TYPE RANGE OF zct_avoid_hardcd-hardcode_key, 



      lrs_key     LIKE LINE OF  lro_key, 

      lt_hardcd   TYPE TABLE OF lty_hardcd, 

      lt_but100   TYPE TABLE OF lty_but100, 

      ls_hardcd   TYPE          lty_hardcd, 

      ls_bapiret2 TYPE          bapiret2, 

      lv_bp_flag  TYPE          boolean, 

      lv_partner  TYPE          but000-partner. 

 

 

    FIELD-SYMBOLS: <lfs_partners> TYPE bapiparnr. 

 

*   Clear the ITAB's. 

    REFRESH : lt_hardcd[],lt_but100[]. 

 

 

*   Check the Partners ITAB is initial or not. 

    CHECK ct_order_partners[] IS NOT INITIAL. 

 

*   Alpha input conversion for customer number 

    LOOP AT  ct_order_partners ASSIGNING  <lfs_partners> . 

      IF <lfs_partners>-partn_numb IS NOT INITIAL. 

        CALL FUNCTION 'CONVERSION_EXIT_ALPHA_INPUT' 

          EXPORTING 

            input  = <lfs_partners>-partn_numb 

          IMPORTING 

            output = <lfs_partners>-partn_numb. 

      ENDIF. 

    ENDLOOP. 

 



*   Prepare the Range table. 

    CLEAR  : lrs_key ,lro_key. 

    lrs_key-sign   = 'I'. 

    lrs_key-option = 'EQ'. 

    lrs_key-low    =  lc_intg_user . 

    APPEND lrs_key TO  lro_key. 

 

    CLEAR  : lrs_key. 

    lrs_key-sign   = 'I'. 

    lrs_key-option = 'EQ'. 

    lrs_key-low    =  lc_def_partner. 

    APPEND lrs_key TO  lro_key. 

 

 

    IF lro_key[] IS NOT INITIAL. 

*     Read  the entries from Hard code table. 

      SELECT    hardcode_key 

                hardcode_value FROM zct_avoid_hardcd 

                               INTO TABLE  lt_hardcd 

                               WHERE  hardcode_key  IN lro_key. 

      IF sy-subrc EQ 0. 

        SORT  lt_hardcd BY  hardcode_key  ASCENDING. 

      ENDIF. 

    ENDIF. 

 

    CLEAR  :  ls_hardcd. 

*   Check if the User is Integration User or not. 

    READ TABLE  lt_hardcd INTO   ls_hardcd WITH KEY hardcode_key = lc_intg_user   BINARY SEARCH. 

    IF sy-subrc NE 0 OR  ls_hardcd-hardcode_value NE  sy-uname   ##USER_OK. 



*     If User is not a Integration user, exit from method. 

      RETURN. 

    ENDIF. 

 

*   Read the partner roles. 

    SELECT   partner 

             rltyp   FROM but100 

                     INTO TABLE lt_but100 

                     FOR ALL ENTRIES IN  ct_order_partners 

                     WHERE partner = ct_order_partners-partn_numb 

                     AND  rltyp  =  lc_cust_role.  "Customer 

    IF sy-subrc EQ 0. 

*     Sort the ITAB. 

      SORT lt_but100 BY partner ASCENDING. 

    ENDIF. 

 

    CLEAR  :lv_bp_flag, ls_bapiret2,  lv_partner , ls_hardcd. 

*   Read the Defult partner from Hardcode table. 

    READ TABLE  lt_hardcd INTO   ls_hardcd WITH KEY hardcode_key = lc_def_partner   BINARY SEARCH. 

    IF sy-subrc EQ 0. 

      lv_partner = ls_hardcd-hardcode_value. 

      UNPACK   lv_partner TO  lv_partner. 

    ELSE. 

      lv_bp_flag =  abap_true. 

    ENDIF. 

 

*   Loop on all Partners. 

    LOOP AT  ct_order_partners ASSIGNING  <lfs_partners> . 

 



*     Check partner have Customer role or not. If not replace with defult partner. 

      READ TABLE lt_but100 WITH KEY partner =  <lfs_partners>-partn_numb  BINARY SEARCH 

TRANSPORTING NO FIELDS. 

      IF sy-subrc NE  0. 

        <lfs_partners>-partn_numb =  lv_partner. 

      ELSE. 

        lv_bp_flag =  abap_false. 

      ENDIF. 

    ENDLOOP. 

 

*   Throw exception and return message in the case if partner not exist in the system and Dummy 

customer is not maintained 

    IF lv_bp_flag =  abap_true. 

 

*     Add the message, if Dummy customer not maintained in table "ZCT_AVOID_HARDCD". 

      CLEAR  :ls_bapiret2. 

 

      ls_bapiret2-type       =  'E'. 

      ls_bapiret2-id         =  lc_mesg_cls. 

      ls_bapiret2-number     =  '000'. 

      ls_bapiret2-message_v1 =  TEXT-001. 

 

*     Get the message text 

      MESSAGE ID  lc_mesg_cls  TYPE ls_bapiret2-type  NUMBER  ls_bapiret2-number 

           WITH  ls_bapiret2-message_v1 INTO ls_bapiret2-message. 

      APPEND ls_bapiret2 TO ct_return. 

 

*     Raise the exception 

      RAISE abort. 



    ENDIF. 

 

  ENDMETHOD. 


